
Abstract

Preference handling and optimization are indispensable means for addressing non-trivial applications
in Answer Set Programming (ASP). However, their implementation becomes difficult whenever they
bring about a significant increase in computational complexity. As a consequence, existing ASP
systems do not offer complex optimization capacities, supporting, for instance, inclusion-based min-
imization or Pareto efficiency. Rather, such complex criteria are typically addressed by resorting to
dedicated modeling techniques, like saturation. Unlike the ease of common ASP modeling, however,
these techniques are rather involved and hardly usable by ASP laymen. We address this problem by
developing a general implementation technique by means of meta-programming, thus reusing exist-
ing ASP systems to capture various forms of qualitative preferences among answer sets. In this way,
complex preferences and optimization capacities become readily available for ASP applications.

1 Introduction

Preferences are often an indispensable means in modeling since they allow for identifying
preferred solutions among all feasible ones. Accordingly, many forms of preferences have
already found their way into systems for Answer Set Programming (ASP; (Baral 2003)).
For instance, smodels provides optimization statements for expressing cost functions on
sets of weighted literals (Simons et al. 2002), and dlv (Leone et al. 2006) offers weak con-
straints for the same purpose. Further approaches (Delgrande et al. 2003; Eiter et al. 2003)
allow for expressing various types of preferences among rules. Unlike this, no readily ap-
plicable implementation techniques are available for qualitative preferences among answer
sets, like inclusion minimality, Pareto-based preferences as used in (Sakama and Inoue
2000; Brewka et al. 2004), or more complex combinations as proposed in (Brewka 2004).
This shortcoming is due to their higher expressiveness leading to a significant increase in
computational complexity, lifting decision problems (for normal logic programs) from the
first to the second level of the polynomial time hierarchy (cf. (Garey and Johnson 1979)).
Roughly speaking, preferences among answer sets combine an NP with a coNP problem.
The first one defines feasible solutions, while the second one ensures that there are no better
solutions according to the preferences at hand. For implementing such problems, Eiter and
Gottlob invented in (1995) the saturation technique, using the elevated complexity of dis-
junctive logic programming. In stark contrast to the ease of common ASP modeling (e.g.,

strategic companies can be “naturally” encoded (Leone et al. 2006) in disjunctive ASP),
however, the saturation technique is rather involved and hardly usable by ASP laymen.

For taking this burden of intricate modeling off the user, we propose a general,
saturation-based implementation technique capturing various forms of qualitative prefer-
ences among answer sets. This is driven by the desire to guarantee immediate availability
and thus to stay within the realm of ASP rather than to build separate (imperative) com-
ponents. To this end, we take advantage of recent advances in ASP grounding technology,
admitting an easy use of meta-modeling techniques. The idea is to reinterpret existing opti-
mization statements in order to express complex preferences among answer sets. While, for
instance in smodels, the meaning of #minimize is to compute answer sets incurring min-
imum costs, we may alternatively use it for selecting inclusion-minimal ones. In contrast
to the identification of minimal models, investigated by Janhunen and Oikarinen in (2004;
2008), a major challenge lies in guaranteeing the stability property of implicit counterex-
amples, which must be more preferred answer sets rather than (arbitrary) models. For this
purpose, we develop a refined meta-program qualifying answer sets as viable counterex-
amples. Unlike the approach of Eiter and Polleres (2006), our encoding avoids “guessing”
a level mapping to describe the formation of a counterexample, but directly denies mod-
els for which there is no such construction. Notably, our meta-programs apply to (reified)
extended logic programs (Simons et al. 2002), possibly including choice rules and #sum

constraints, and we are unaware of any existing meta-encoding of their answer sets, neither
as candidates nor as counterexamples refuting optimality.

2 Background

We consider extended logic programs (Simons et al. 2002) allowing for (proper) disjunc-
tions in heads of rules (Gelfond and Lifschitz 1991). A rule r is of the following form:

H ← B1, . . . , Bm,∼Bm+1, . . . ,∼Bn.

By head(r) = H and body(r) = {B1, . . . , Bm,∼Bm+1, . . . ,∼Bn}, we denote the head
and the body of r, respectively, where “∼” stands for default negation. The head H is a
disjunction a1 ∨ · · · ∨ ak over atoms a1, . . . , ak, belonging to some alphabetA, or a #sum

constraint L#sum[`1 = w1, . . . , `k = wk]U . In the latter, `i = ai or `i = ∼ai is a literal
and wi a non-negative integer weight for ai ∈ A and 1 ≤ i ≤ k; L and U are integers
providing a lower and an upper bound. Either or both of L and U can be omitted, in which
case they are identified with the (trivial) bounds 0 and∞, respectively. A rule r such that
head(r) = ⊥ (H is the empty disjunction) is an integrity constraint. Each body component
Bi is either an atom or a #sum constraint for 1 ≤ i ≤ n. If body(r) = ∅, r is called a fact,
and we skip “←” when writing facts below. For a set {B1, . . . , Bm,∼Bm+1, . . . ,∼Bn}, a
disjunction a1∨· · ·∨ak, and a #sum constraint L#sum[`1 = w1, . . . , `k = wk]U , we let
{B1, . . . , Bm,∼Bm+1, . . . ,∼Bn}+ = {B1, . . . , Bm}, (a1 ∨ · · · ∨ ak)+ = {a1, . . . , ak},
and (L#sum[`1 = w1, . . . , `k = wk]U)+ = [`i = wi | 1 ≤ i ≤ k, `i ∈ A]. Note that the
elements of a #sum constraint form a multiset, possibly containing duplicates. For some
S = {a1, . . . , ak} or S = [a1 = w1, . . . , ak = wk], we define atom(S) = {a1, . . . , ak}.

A (Herbrand) interpretation is represented by the set X ⊆ A of its entailed atoms. The
satisfaction relation “|=” on rules r is inductively defined as follows:

• X |= ∼B if X 6|= B,
• X |= (a1 ∨ · · · ∨ ak) if {a1, . . . , ak} ∩X 6= ∅,
• X |= (L#sum[`1 = w1, . . . , `k = wk]U) if L ≤

∑
1≤i≤k,X|=`i

wi ≤ U ,
• X |= body(r) if X |= ` for all ` ∈ body(r), and
• X |= r if X |= head(r) or X 6|= body(r).

A logic program Π is a set of rules r, and X is a model of Π if X |= r for every r ∈ Π.
The reduct of the head H of a rule r wrt X is HX = {a1 ∨ · · · ∨ ak} if H = a1 ∨ · · · ∨ ak,
and HX = atom(H+) ∩X if H = L#sum[`1 = w1, . . . , `k = wk]U . Furthermore, the
reduct of some (positive) body element B ∈ body(r)+ is BX = B if B ∈ A, and BX =(
L−

∑
1≤i≤k,`i=∼ai,ai /∈X wi

)
#sumB+ if B = L#sum[`1 = w1, . . . , `k = wk]U . The

reduct of Π wrt X is the following logic program:

ΠX =
{
H ← BX

1 , . . . , BX
m |

r ∈ Π, X |= body(r), H ∈ head(r)X , body(r)+ = {B1, . . . , Bm}
}
.

That is, for all rules r ∈ Π whose bodies are satisfied wrt X , the reduct is obtained by
replacing #sum constraints in heads with individual atoms belonging to X and by elimi-
nating negative components in bodies, where lower bounds of residual #sum constraints
(with trivial upper bounds) are reduced accordingly. Finally, X is an answer set of Π if X
is a model of Π such that no proper subset of X is a model of ΠX . In view of the latter
condition, note that an answer set is a minimal model of its own reduct.

The definition of answer sets provided above applies to logic programs containing ex-
tended constructs (#sum constraints) under “choice semantics” (Simons et al. 2002), while
additionally allowing for disjunctions under minimal-model semantics (wrt a reduct). We
use these features to embed extended constructs of an object program into a disjunctive
meta-program, so that their combination yields optimal answer sets of the object program.
To this end, we reinterpret #minimize statements of the following form:

#minimize[`1 = w1@J1, . . . , `k = wk@Jk]. (1)

Like with #sum constraints, every `i is a literal and every wi an integer weight for
1 ≤ i ≤ k, while Ji additionally provides an integer priority level.1 Priorities allow
for representing a sequence of lexicographically ordered #minimize objectives, where
greater levels are more significant than smaller ones. By default, a #minimize statement
distinguishes optimal answer sets of a program Π in the following way. For any X ⊆ A
and integer J , let ΣX

J denote the sum of weights w over all occurrences of weighted literals
` = w@J in (1) such that X |= `. An answer set X of Π is dominated if there is an answer
set Y of Π such that ΣY

J < ΣX
J and ΣY

J′ = ΣX
J′ for all J ′ > J , and optimal otherwise.

In the following, we assume that every logic program is accompanied with one (possibly
empty) #minimize statement of the form (1). Instead of the default semantics, we con-
sider Pareto efficiency wrt priority levels J , weights w, and several distinct optimization

1 Explicit priority levels are supported in recent versions of the grounder gringo (Gebser et al.). This avoids
a dependency of priorities on input order, which is considered by lparse (Syrjänen) if several #minimize
statements are provided. Priority levels are also supported by dlv (Leone et al. 2006) in weak constraints.
Furthermore, we admit negative weights in #minimize statements, where they cannot raise semantic problems
(cf. (Ferraris 2005)) going along with the rewriting of #sum constraints suggested in (Simons et al. 2002).

criteria. In view of this, we use levels for inducing a lexicographic order, while weights are
used for grouping literals (rather than summation). Pareto improvement then builds upon
a two-dimensional structure of orderings among answer sets, induced by J and w. In turn,
each such pairing is associated with some of the following orderings. By Y ≤w

J X , we
denote that the cardinality of the multiset of occurrences of ` = w@J in (1) such that
Y |= ` is not greater than the one of the corresponding multiset for X |= `. Furthermore,
we write Y ⊆w

J X if, for any weighted literal ` = w@J occurring in (1), Y |= ` implies
X |= `. Finally, we denote by Y �w

J X that Y is preferable to X (Sakama and Inoue 2000)
according to a (given) preference relation � among literals ` such that ` = w@J occurs
in (1). Given a logic program Π and a collection M of relations of the form �wJ for priority
levels J , weights w, and � ∈ {≤,⊆,�}, an answer set Y of Π dominates an answer set X
of Π wrt M if there are a priority level J and a weight w such that X �wJ Y does not hold
for �wJ ∈M , while Y �w′

J′ X holds for all �w′

J′ ∈M where J ′ ≥ J . In turn, an answer set X
of Π is optimal wrt M if there is no answer set Y of Π that dominates X wrt M .

As an example, consider the following program, referred to by Π0:

1 {p, t} ← 1 {r, s,∼t} 2. (2)

{q, r} 1 ← 1 {p, t}. (3)

s ← ∼q,∼r. (4)

This program has five answer sets, viz. {p, q}, {p, r}, {p, s}, {p, s, t}, and {s, t}. (Sets
{a1, . . . , ak} in (2) and (3) are used as shorthands for #sum[a1 = 1, . . . , ak = 1].) In
addition, let Π1 denote the union of Π0 with the following #minimize statement:

#minimize[p = 1@1, q = 1@1, r = 1@1, s = 1@1]. (5)

This statement specifies that all atoms of Π0 except for t are subject to minimization. Pass-
ing Π1 to gringo and an answer set solver like smodels yields the single≤1

1-minimal answer
set {s, t}. Note, however, that Π0 has three⊆1

1-minimal answer sets, namely {p, q}, {p, r},
and {s, t}. They cannot be computed directly from Π1 via any available ASP system.

We implement the complex optimization criteria described above by meta-interpretation
in disjunctive ASP. For transparency, we provide meta-programs as true ASP code in the
first-order input language of gringo (Gebser et al.), including not and | as tokens for ∼
and ∨, respectively, as well as {a1,. . .,ak} as shorthand for #sum[a1=1,. . .,ak=1].
Further constructs are informally introduced by need in the remainder of this paper. Note
that our (disjunctive) meta-programs apply to an extended object program that does not
include proper disjunctions (over more than one atom). Unless stated otherwise, we below
use the term extended program to refer to a logic program without proper disjunctions.

3 Basic Meta-Modeling

For reinterpreting #minimize statements by means of ASP, we take advantage of recent
advances in ASP grounding, admitting an easy use of meta-modeling techniques. To be
precise, we rely upon the unrestricted usage of function symbols and program reification as
provided by gringo (Gebser et al.). The latter allows for turning an input program along with
a #minimize statement into facts representing the structure of their ground instantiation
via fixed sets of predicates and function symbols.

1 rule(pos(sum(1,0,2)),pos(conjunction(0))). % 1 { p, t } :- 1 { r, s, not t } 2.
2 wlist(0,0,pos(atom(p)),1). wlist(0,1,pos(atom(t)),1).
3 set(0,pos(sum(1,1,2))).
4 wlist(1,0,pos(atom(r)),1). wlist(1,1,pos(atom(s)),1). wlist(1,2,neg(atom(t)),1).

6 rule(pos(sum(0,2,1)),pos(conjunction(1))). % { q, r } 1 :- 1 { p, t }.
7 wlist(2,0,pos(atom(q)),1). wlist(2,1,pos(atom(r)),1).
8 set(1,pos(sum(1,0,2))).

10 rule(pos(atom(s)),pos(conjunction(2))). % s :- not q, not r.
11 set(2,neg(atom(q))). set(2,neg(atom(r))).

13 scc(0,pos(atom(p))). scc(0,pos(atom(r))). scc(0,pos(atom(t))).
14 scc(0,pos(conjunction(0))). scc(0,pos(sum(1,1,2))).
15 scc(0,pos(conjunction(1))). scc(0,pos(sum(1,0,2))).

17 minimize(1,3). % #minimize [ p = 1 @ 1, q = 1 @ 1, r = 1 @ 1, s = 1 @ 1 ].
18 wlist(3,0,pos(atom(p)),1). wlist(3,1,pos(atom(q)),1).
19 wlist(3,2,pos(atom(r)),1). wlist(3,3,pos(atom(s)),1).

Listing 1. Facts describing a reified extended logic program.

For illustrating the format output by gringo, consider the facts in Line 1–15 of Listing 1,
obtained by calling gringo with option --reify on program Π0. Let us detail the repre-
sentation of the rule in (2) inducing the facts in Line 1–4. The predicate rule/2 is used to
link the rule head and body. By convention, both are positive rule elements, as indicated via
the functor pos/1. Furthermore, the term sum(1,0,2) tells us that the head is a #sum

constraint with lower bound 1 and (trivial) upper bound 2 over a list labeled 0 of weighted
literals. In fact, the included literals are provided via the facts over wlist/4 given in
Line 2, whose first arguments are 0. While the second arguments, 0 and 1, are simply
indexes (enabling the representation of duplicates in multisets), the third ones provide lit-
erals, p and t, each having the (default) weight 1, as given in the fourth arguments. Again
by convention, the body of each rule is a conjunction, where the term conjunction(0)
in Line 1 refers to the set labeled 0. Its single element, a positive #sum constraint with
lower bound 1 and upper bound 2 over a list labeled 1, is provided by the fact in Line 3.
The corresponding weighted literals are described by the facts in Line 4; observe that the
negative literal not t is represented in terms of the functor neg/1, applied to atom(t).
The rules in (3) and (4) are represented analogously in Line 6–8 and 10–11, respectively. It
is still interesting to note that recurrences of lists of weighted literals (and sets) can reuse
labels introduced before, as done in Line 8 by referring to 0. In fact, gringo identifies repe-
titions of structural entities and reuses labels. In addition to the rules of Π0, the elements of
non-trivial strongly connected components of its positive dependency graph (cf. (6) below)
are provided in Line 13–15. Albeit their usage is explained in the next section, note already
that the members of the only such component, labeled 0, include atoms as well as (posi-
tive) body elements, i.e., conjunctions and #sum constraints, connecting the component.
Indeed, the existence of facts over scc/2 tells us that Π0 is not tight (cf. (Fages 1994)).

Now, we may compute all five answer sets of Π0 (given in p0.lp) by combining the
facts in Line 1–15 of Listing 1 with the basic meta-program in Listing 2 (meta.lp):2

gringo --reify p0.lp | gringo meta.lp - | clasp 0

2 Following Unix customs, the minus symbol “-” stands for the output of “gringo --reify p0.lp.”

1 % extract rule elements

3 litb(B) :- rule(_,B).
4 litb(E) :- litb(pos(conjunction(S))), set(S,E).
5 litb(E) :- eleb(sum(_,S,_)), wlist(S,_,E,_).

7 eleb(P) :- litb(pos(P)).
8 eleb(N) :- litb(neg(N)).

10 elem(E) :- eleb(E).
11 elem(E) :- rule(pos(E),_).
12 elem(P) :- rule(pos(sum(_,S,_)),_), wlist(S,_,pos(P),_).
13 elem(N) :- rule(pos(sum(_,S,_)),_), wlist(S,_,neg(N),_).

15 % generate answer set from reified rules

17 hold(conjunction(S)) :- eleb(conjunction(S)),
18 hold(P) : set(S,pos(P)),
19 not hold(N) : set(S,neg(N)).
20 hold(sum(L,S,U)) :- eleb(sum(L,S,U)),
21 L #sum [ hold(P) = W : wlist(S,Q,pos(P),W),
22 not hold(N) = W : wlist(S,Q,neg(N),W) ] U.

24 hold(atom(A)) :- rule(pos(atom(A)), pos(B)), hold(B).
25 L #sum [ hold(P) = W : wlist(S,Q,pos(P),W),
26 not hold(N) = W : wlist(S,Q,neg(N),W) ] U
27 :- rule(pos(sum(L,S,U)),pos(B)), hold(B).
28 :- rule(pos(false), pos(B)), hold(B).

30 % project output to atoms of answer set

32 #hide. #show hold(atom(A)).

Listing 2. Basic meta-program (meta.lp) for reified extended logic programs.

Each answer set of the meta-program applied to a reified program corresponds to an an-
swer set of the reified program. More precisely, a set X of atoms is an answer set of
the reified program iff the meta-program yields an answer set Y such that X = {a |
hold(atom(a)) ∈ Y }, e.g., hold(atom(q)) stands for q. As indicated in the com-
ments (preceded by %), our meta-program consists of three parts. Among the rule elements
extracted in Line 3–13, only those occurring within bodies, identified via eleb/1, are
relevant to the generation of answer sets specified in Line 17–28. (Additional head ele-
ments, given by elem/1, are of interest in the next section.) In fact, answer set generation
follows the structure of reified programs, identifying conjunctions and #sum constraints
that hold3 to further derive atoms occurring in rule heads, either singular or within #sum

constraints (cf. Line 24–27). Line 28 deals with integrity constraints represented via the
constant false in heads of reified rules. The last part in Line 32 restricts the output of the
meta-program’s answer sets to the representations of original input atoms.

Finally, note that meta.lp does not inspect facts representing a reified #minimize

statement, such as the ones in Line 17–19 of Listing 1 stemming from the statement in (5).
Such facts over minimize/2 provide a priority level as the first argument and the label of
a list of weighted literals, like the ones referred to from within terms of functor sum/3, as
the second argument. Rather than simply mirroring the standard meaning of #minimize

statements (by encoding them analogously to rules; cf. Line 17–28 of Listing 2), we support

3 The “:” connective expands to the list of all instances of its left-hand side such that corresponding instances
of literals on the right-hand side hold (cf. (Syrjänen) and (Gebser et al.)).

flexible customizations. In fact, the next section presents our meta-programs implementing
preference relations and Pareto efficiency, as described in the background.

4 Advanced Meta-Modeling

Given the reification of extended logic programs and the encoding of their answer sets in
meta.lp, our approach to complex optimization is based on the idea that an answer set
generated via meta.lp is optimal (and thus acceptable) only if it is not dominated by any
other answer set. For implementing our approach, we exploit the capabilities of disjunctive
ASP to compactly represent the space of all potential counterexamples, viz. answer sets
dominating a candidate answer set at hand. To this end, we encode the subtasks of

1. guessing an answer set as a potential counterexample and
2. verifying that the counterexample dominates a candidate answer set.

A candidate answer set passes both phases if it turns out to be infeasible to guess a coun-
terexample that dominates it. For expressing the non-existence of counterexamples, we
make use of an error-indicating atom bot and saturation (Eiter and Gottlob 1995), deriv-
ing all atoms representing the space of counterexamples from bot. Since the semantics
of disjunctive ASP is based on minimization, saturation makes sure that bot is derived
only if it is inevitable, i.e., if it is impossible to construct a counterexample. However, via
an integrity constraint, we can stipulate bot (and thus the non-existence of counterexam-
ples) to hold, yet without providing any derivation of bot. In view of such a constraint
and saturation, a successful candidate answer set is accompanied by all atoms representing
counterexamples. Given that the reduct drops negative literals, the necessity that all atoms
representing counterexamples are true implies that we cannot use their default negation in
any meaningful way. Hence, we below encode potential counterexamples, i.e., answer sets
of extended programs, and (non-)dominance of a candidate answer set in disjunctive ASP
without taking advantage of default negation (used in meta.lp).

For encoding the first subtask of guessing a counterexample, we rely on a characteriza-
tion of answer sets in terms of an immediate consequence operator T (cf. (Lloyd 1987)),
defined as follows for a logic program Π and a set X ⊆ A of atoms: TΠ(X) = {head(r) |
r ∈ Π, X |= body(r)}. Furthermore, an iterative version of T can be defined in the fol-
lowing way: T 0

Π(X) = X and T i+1
Π (X) = T i

Π(X) ∪ TΠ(T i
Π(X)). In the context of an

extended program Π, possibly including choice rules, default negation, and upper bounds
of weight constraints, we are interested in the least fixpoint of T applied wrt the reduct ΠX .
Since a fixpoint is reached in at most |atom(Π)| applications of T , where atom(Π) ⊆ A
denotes the set of atoms occurring in Π, the least fixpoint is given by T |atom(Π)|

ΠX (∅). As
pointed out in (Liu and You 2010), a model X of an extended program Π is an answer
set of Π iff T |atom(Π)|

ΠX (∅) = X . Furthermore, Liu and You (2010) show that X violates
the loop formula of some atom or loop if X is a model, but not an answer set of Π. This
property motivates a “localization” of T on the basis of (circular) positive dependencies.

The (positive) dependency graph of an extended program Π is given by the following
pair of nodes and directed edges:(

atom(Π), {(a, b) | r ∈ Π, a ∈ atom(head(r)+), B ∈ body(r)+, b ∈ atom(B+)}
)
. (6)



8 Martin Gebser and Roland Kaminski and Torsten Schaub

A strongly connected component (SCC) is a maximal subgraph of the dependency graph
of Π such that all nodes are pairwisely connected via paths. An SCC is trivial if it does not
contain any edge, and non-trivial otherwise. Note that the SCCs of the dependency graph
of Π induce a partition of atom(Π) such that every atom and every loop of Π is contained
in some part. Hence, we can make use of the partition to apply T separately to each part.

Proposition 1
Let Π be an extended logic program, C1, . . . , Ck be the sets of atoms belonging to the
SCCs of the dependency graph of Π, and X ⊆ atom(Π).

Then, we have that T |atom(Π)|
ΠX (∅) = X iff

⋃
1≤j≤k(T |Cj |

ΠX (X \ Cj) ∩ Cj) = X .

Proof (Sketch)
It is clear that T |atom(Π)|

ΠX (∅) = X implies T |Cj |
ΠX (X \ Cj) ∩ Cj = X ∩ Cj for every

1 ≤ j ≤ k. Hence, we only need to show that
⋃

1≤j≤k(T |Cj |
ΠX (X \Cj)∩Cj) = X implies

T |atom(Π)|
ΠX (∅) = X . To this end, assume that

⋃
1≤j≤k(T |Cj |

ΠX (X \ Cj) ∩ Cj) = X and
that C1, . . . , Ck are topologically ordered such that the dependency graph of Π does not
contain any edge from an atom in Cj to atoms in Cj+1 ∪ · · · ∪ Ck for 1 ≤ j ≤ k. That is,
atoms in Cj+1 ∪ · · · ∪ Ck do not occur in rules r ∈ ΠX such that head(r) ∈ Cj , so that
T |Cj |

ΠX (X \ Cj) ∩ Cj = T |Cj |
ΠX (X ∩ (C1 ∪ · · · ∪ Cj−1)) ∩ Cj for 1 ≤ j ≤ k. From X =

(X∩C1)∪· · ·∪(X∩Ck) = (T |C1|
ΠX (∅)∩C1)∪· · ·∪(T |Ck|

ΠX (X∩(C1∪· · ·∪Ck−1))∩Ck) ⊆
T |C1|+···+|Ck|

ΠX (∅) = T |atom(Π)|
ΠX (∅) ⊆ X , we then conclude that T |atom(Π)|

ΠX (∅) = X .

For illustration, reconsider Π0 in (2)–(4) and its dependency graph, looking as follows:

p tr

s

q

q
i

q
iPPPPPq

�����)

�����)
PPPPPq

Observe that the sets {s}, {p, r, t}, and {q} of atoms belong to SCCs (ordered topologi-
cally). Furthermore, let us take the following reducts into account:

Π
{p,r}
0 =

{
p← 0 #sum[r = 1, s = 1].

r ← 1 #sum[p = 1, t = 1].

}
Π
{r,t}
0 =

{
t← 1 #sum[r = 1, s = 1].

r ← 1 #sum[p = 1, t = 1].

}
We have that T 1

Π
{p,r}
0

(∅) = {p} and T 2

Π
{p,r}
0

(∅) = {p} ∪ T
Π

{p,r}
0

({p}) = {p, r} =

T 3

Π
{p,r}
0

(∅). Along with T 1

Π
{p,r}
0

({p, r}) ∩ {s} = T 1

Π
{p,r}
0

({p, r}) ∩ {q} = ∅, we obtain

(T 1

Π
{p,r}
0

({p, r}) ∩ {s}) ∪ (T 3

Π
{p,r}
0

(∅) ∩ {p, r, t}) ∪ (T 1

Π
{p,r}
0

({p, r}) ∩ {q}) = {p, r}. In

view of Proposition 1, this confirms that the model {p, r} of Π0 is an answer set of Π0. On
the other hand, (T 1

Π
{r,t}
0

({r, t})∩{s})∪(T 3

Π
{r,t}
0

(∅)∩{p, r, t})∪(T 1

Π
{r,t}
0

({r, t})∩{q}) = ∅
yields that the model {r, t} of Π0 is not an answer set of Π0.

In a nutshell, our encoding of answer sets (as counterexamples) in disjunctive ASP com-
bines the following parts:

1. guessing an interpretation,
2. deriving the error-indicating atom bot if the interpretation is not a supported model

(where each true atom occurs positively in the head of some rule whose body holds),

3. deriving bot if the true atoms of some non-trivial SCC are not acyclicly derivable
(checked via determining the complement of a fixpoint of T ), and

4. saturating interpretations that do not correspond to answer sets by deriving all truth
assignments (for atoms) from bot.

Note that the third part, checking acyclic derivability, concentrates on atoms of non-trivial
SCCs, while checking support in the second part is already sufficient for trivial SCCs.

The meta-program in Listing 3 implements the sketched idea. In the following, we con-
centrate on describing its crucial features. For evaluating support, the meta-rules in Line 3
and 4 collect atoms having a positive occurrence in the head of a rule along with the
rule’s body. Note that, for atoms contained in a #sum constraint in the head, the asso-
ciated bounds and weights are inessential in the context of support. On the other hand, the
meta-rule in Line 6 sums the weights of all literals in a #sum constraint; this is needed
to evaluate bounds in the sequel, where (non-reified) default negation and upper bounds
(acting negatively) are inapplicable in view of saturation.

The meta-rules in Line 10–29 generate an interpretation by guessing some truth value
for each atom (Line 10) and evaluating further constructs occurring in a reified program
accordingly (Line 12–29). While the special constant false (used as head of integrity
constraints) holds in no interpretation (fail(false) is a fact) and the evaluation of
conjunctions is straightforward, more care is required for evaluating #sum constraints. For
instance, the case that a #sum constraint holds is in the meta-rule in Line 19–23 identified
via sufficiently many literals that hold to achieve the lower bound L and also sufficiently
many literals that do not hold to fill the gap between the upper bound U and the sum T of
all weights. Note that the latter condition is encoded by the lower bound T-U, rather than
taking U as an upper bound (as done in meta.lp). The complementary cases that a #sum

constraint does not hold are described in the same manner in Line 24–29, where the lower
bound T-L+1 (or U+1) for weights of literals that do not hold (or hold) is used to indicate
a violated lower (or upper) bound of the reified #sum constraint.

Given an interpretation of atoms and the corresponding truth values of further constructs
in an extended program, the meta-rules in Line 33 and 34 are used to derive bot if the
interpretation does not provide us with a supported model. To avoid such a derivation
of bot, every rule of the reified program must be satisfied, and every true atom must have
a positive occurrence in the head of some rule whose body holds.

It remains to check the acyclic derivability of atoms belonging to non-trivial SCCs. To
this end, the meta-rule in Line 38 determines the number Z of atoms in an SCC labeled C
as the maximum step at which a fixpoint of T , applied locally to C, is reached. Further-
more, the meta-rule in Line 40–41 derives sccw(A) if the atom referred to by A does not
have a derivation external to C. (Recall that the positive body elements of rules internally
connecting an SCC, i.e., rules contributing the SCC’s edges to the dependency graph, are
marked by facts over scc/2; cf. Listing 1.) The acyclic derivability of atoms indicated by
sccw(A) is of particular interest in the sequel. In fact, our encoding identifies the com-
plement of a fixpoint of T in terms of atoms A for which wait(atom(A),Z) is derived.
To accomplish this, the meta-rule in Line 45 marks all atoms of C as underived at step 0.
As encoded via the meta-rule in Line 46–47, an atom A stays underived at a later step D if
there is no external derivation of A (sccw(A) holds) and the bodies B of all component-

1 % extract supports of atoms and sums of weight lists’ weights

3 supp(atom(A),B) :- rule(pos(atom(A)), pos(B)).
4 supp(atom(A),B) :- rule(pos(sum(_,S,_)),pos(B)), wlist(S,_,pos(atom(A)),_).

6 sum(S,T) :- elem(sum(_,S,_)), T = #sum [ wlist(S,Q,_,W) = W ].

8 % generate interpretation

10 true(atom(A)) | fail(atom(A)) :- elem(atom(A)).

12 fail(false).

14 true(conjunction(S)) :- elem(conjunction(S)),
15 true(P) : set(S,pos(P)), fail(N) : set(S,neg(N)).
16 fail(conjunction(S)) :- elem(conjunction(S)), set(S,pos(P)), fail(P).
17 fail(conjunction(S)) :- elem(conjunction(S)), set(S,neg(N)), true(N).

19 true(sum(L,S,U)) :- elem(sum(L,S,U)), sum(S,T),
20 L #sum [ true(P) = W : wlist(S,Q,pos(P),W),
21 fail(N) = W : wlist(S,Q,neg(N),W) ],
22 T-U #sum [ fail(P) = W : wlist(S,Q,pos(P),W),
23 true(N) = W : wlist(S,Q,neg(N),W) ].
24 fail(sum(L,S,U)) :- elem(sum(L,S,U)), sum(S,T),
25 T-L+1 #sum [ fail(P) = W : wlist(S,Q,pos(P),W),
26 true(N) = W : wlist(S,Q,neg(N),W) ].
27 fail(sum(L,S,U)) :- elem(sum(L,S,U)),
28 U+1 #sum [ true(P) = W : wlist(S,Q,pos(P),W),
29 fail(N) = W : wlist(S,Q,neg(N),W) ].

31 % verify supported model properties

33 bot :- rule(pos(H),pos(B)), true(B), fail(H).
34 bot :- true(atom(A)), fail(B) : supp(atom(A),B).

36 % verify acyclic derivability

38 step(C,Z) :- scc(C,_), Z = #sum [ scc(C,pos(atom(A))) ].

40 sccw(A) :- scc(C,pos(atom(A))),
41 fail(B) : supp(atom(A),B) : not scc(C,pos(B)).

43 wait(E,D-1) :- scc(C,pos(E)), fail(E), step(C,Z), D = 1..Z.

45 wait(atom(A),0) :- scc(C,pos(atom(A))).
46 wait(atom(A),D) :- scc(C,pos(atom(A))), sccw(A), step(C,Z), D = 1..Z,
47 wait(B,D-1) : supp(atom(A),B) : scc(C,pos(B)).

49 wait(sum(L,S,U),D-1) :- scc(C,pos(sum(L,S,U))), sum(S,T), step(C,Z), D = 1..Z,
50 T-L+1 #sum [ fail(P) = W : wlist(S,Q,pos(P),W) : not scc(C,pos(P)),
51 wait(P,D-1) = W : wlist(S,Q,pos(P),W) : scc(C,pos(P)),
52 true(N) = W : wlist(S,Q,neg(N),W) ].

54 wait(conjunction(S),D-1) :- scc(C,pos(conjunction(S))), set(S,pos(P)),
55 scc(C,pos(P)), wait(P,D-1), step(C,Z), D = 1..Z.

57 bot :- scc(C,pos(atom(A))), true(atom(A)), wait(atom(A),Z), step(C,Z).

59 % saturate interpretations that are not answer sets

61 true(atom(A)) :- elem(atom(A)), bot.
62 fail(atom(A)) :- elem(atom(A)), bot.

Listing 3. Disjunctive meta-program (metaD.lp) for reified extended logic programs.

internal supports of A are yet underived at step D-1 (wait(B,D-1) holds). The latter is
checked via the meta-rules in Line 49–52 and 54–55, respectively. The former applies to
#sum constraints and identifies cases where the weights of literals that do not hold along

1 step(0,3). sccw(p). sccw(r). sccw(t).

3 wait(atom(p), 0;1;2) :- fail(atom(p)).
4 wait(atom(r), 0;1;2) :- fail(atom(r)).
5 wait(atom(t), 0;1;2) :- fail(atom(t)).
6 wait(conjunction(0),0;1;2) :- fail(conjunction(0)).
7 wait(sum(1,1,2), 0;1;2) :- fail(sum(1,1,2)).
8 wait(conjunction(1),0;1;2) :- fail(conjunction(1)).
9 wait(sum(1,0,2), 0;1;2) :- fail(sum(1,0,2)).

11 wait(atom(p),0).
12 wait(atom(p),1) :- sccw(p), wait(conjunction(0),0).
13 wait(atom(p),2) :- sccw(p), wait(conjunction(0),1).
14 wait(atom(p),3) :- sccw(p), wait(conjunction(0),2).

16 wait(atom(r),0).
17 wait(atom(r),1) :- sccw(r), wait(conjunction(1),0).
18 wait(atom(r),2) :- sccw(r), wait(conjunction(1),1).
19 wait(atom(r),3) :- sccw(r), wait(conjunction(1),2).

21 wait(atom(t),0).
22 wait(atom(t),1) :- sccw(t), wait(conjunction(0),0).
23 wait(atom(t),2) :- sccw(t), wait(conjunction(0),1).
24 wait(atom(t),3) :- sccw(t), wait(conjunction(0),2).

26 wait(sum(1,1,2),0) :- 3 #sum [ fail(atom(s)), wait(atom(r),0), true(atom(t)) ].
27 wait(sum(1,1,2),1) :- 3 #sum [ fail(atom(s)), wait(atom(r),1), true(atom(t)) ].
28 wait(sum(1,1,2),2) :- 3 #sum [ fail(atom(s)), wait(atom(r),2), true(atom(t)) ].

30 wait(sum(1,0,2),0) :- 2 #sum [ wait(atom(p),0), wait(atom(t),0) ].
31 wait(sum(1,0,2),1) :- 2 #sum [ wait(atom(p),1), wait(atom(t),1) ].
32 wait(sum(1,0,2),2) :- 2 #sum [ wait(atom(p),2), wait(atom(t),2) ].

34 wait(conjunction(0),0) :- wait(sum(1,1,2),0).
35 wait(conjunction(0),1) :- wait(sum(1,1,2),1).
36 wait(conjunction(0),2) :- wait(sum(1,1,2),2).

38 wait(conjunction(1),0) :- wait(sum(1,0,2),0).
39 wait(conjunction(1),1) :- wait(sum(1,0,2),1).
40 wait(conjunction(1),2) :- wait(sum(1,0,2),2).

42 bot :- true(atom(p)), wait(atom(p),3).
43 bot :- true(atom(r)), wait(atom(r),3).
44 bot :- true(atom(t)), wait(atom(t),3).

Listing 4. (Simplified) ground rules obtained from meta-rules in Line 38–57 of metaD.lp.

with the ones of yet underived atoms of C exceed T-L, so that the lower bound L is not yet
established. Similarly, the underivability of a conjunction is recognized via a yet underived
positive body element internal to the component C. Also note that the falsity of elements
of C is propagated via the meta-rule in Line 43, so that false atoms, #sum constraints,
and conjunctions do not contribute to derivations of atoms of C. As mentioned above, the
complement of a fixpoint of T contains the atoms A such that wait(atom(A),Z) is
eventually derived. If any such atom A is true, failure to construct an answer set is indi-
cated by deriving bot via the meta-rule in Line 57.

For illustration, consider the ground rules shown in Listing 4, which are ob-
tained for the SCC labeled 0 in Listing 1. For the answer set {p, r} of Π0, rep-
resented by the atoms true(atom(p)), true(atom(r)), fail(atom(q)),
fail(atom(s)), and fail(atom(t)), we have that wait(sum(1,1,2),
0;1;2) and wait(conjunction(0),0;1;2) are underivable via the rules
in Line 26–28 and 34–36, respectively. Thus, wait(atom(p),1;2;3) are not
derived via the rules in Line 12–14, so that wait(sum(1,0,2),1;2) and

wait(conjunction(1),1;2) are underivable in turn via the rules in Line 31–32
and 39–40, respectively. As a consequence, wait(atom(r),2;3) are not derived via
the rules in Line 18–19. We have thus checked that none of the rules in Line 42–44 allows
for deriving bot, which tells us that the true atoms p and r are acyclicly derivable. On the
other hand, for the interpretation {r, t}, given by true(atom(r)), true(atom(t)),
fail(atom(p)), fail(atom(q)), and fail(atom(s)), we have that the atoms
wait(atom(p),0;1;2), wait(sum(1,1,2),0;1;2), wait(sum(1,0,2),
0;1;2), wait(conjunction(0),0;1;2), wait(conjunction(1),0;1;2),
wait(atom(r),0;1;2;3), and wait(atom(t),0;1;2;3) are derived in turn
via the rules in Line 3 and 16–40. From true(atom(r)) and wait(atom(r),3)
as well as true(atom(t)) and wait(atom(t),3), we further derive the error-
indicating atom bot via the rules in Line 43 and 44. This signals that the true atoms r

and t are not acyclicly derivable, so that {r, t} is not an answer set of Π0.
Finally, saturation of interpretations that do not correspond to answer sets is accom-

plished via the meta-rules in Line 61 and 62 of Listing 3. They make sure that bot is in-
cluded in an answer set of the meta-program only if it is inevitable wrt every interpretation.
When considering the encoding part in Listing 3 in isolation, it like meta.lp describes
answer sets of a reified program, and bot is derived only if there is no such answer set.

Our meta-programs meta.lp and metaD.lp in Listing 2 and 3 have not yet consid-
ered facts minimize(J,S) in reified programs, reflecting input #minimize statements.
In fact, complex optimization is addressed by the meta-program metaO.lp, whose first
part is shown in Listing 5. It allows for separate optimization criteria per priority level J
and weight W (in facts wlist(S,Q,E,W)). Particular criteria can be provided via the
user predicate optimize(J,W,O), where the values card, incl, and pref for O
refer to minimality regarding cardinality, inclusion, and preference (Sakama and Inoue
2000), respectively, among the involved literals E. Such criteria are reflected via instances
of cxopt(J,W,O), derived via the rules in Line 7 and 8–9, where card is taken by
default if no criterion is provided by the user. At each priority level J, Pareto improvement
of a counterexample (constructed via the rules in metaD.lp) over all weights W and cri-
teria O such that cxopt(J,W,O) holds is used for deciding whether a candidate answer
set (constructed via the rules in meta.lp) is optimal. To this end, similarity at a priority
level J is indicated by deriving equal(J) from equal(J,W,O) over all instances of
cxopt(J,W,O) via the rule in Line 13. Furthermore, the rules in Line 15–19 are used
to chain successive priority levels, where a greater level J1 is more significant than its
smaller neighbor J2, and to signal whether a priority level J2 is taken into account. The
latter is the case if equal(J1) has been derived at all more significant priority levels J1.
If it turns out that a candidate answer set is not refuted by a dominating counterexam-
ple, we derive bot via the rules in Line 21, 22, and 23: the first rule applies if there
are no optimization criteria at all, the second one checks whether the counterexample is
worse (or incomparable), as indicated by worse(J1) at an inspected priority level J1,
and the third one detects lack of Pareto improvement from equality at the lowest priority
level. Finally, the integrity constraint in Line 27 stipulates bot to hold. Along with satu-
ration (in metaD.lp), this implies that a candidate answer set (constructed via the rules
in meta.lp) is accepted only if there is no dominating counterexample, thus selecting
exactly the optimal answer sets of an input program. The described rules serve the general

1 % extract (complex) optimization criteria per priority level and weight
2 % (relative to user predicate optimize/3; cardinality taken by default;
3 % Pareto improvement over weights used for comparison at a priority level)

5 cxopt(card). cxopt(incl). cxopt(pref).

7 cxopt(J,W,O) :- minimize(J,S), wlist(S,_,_,W), cxopt(O), optimize(J,W,O).
8 cxopt(J,W,card) :- minimize(J,S), wlist(S,_,_,W),
9 not optimize(J,W,O) : optimize(J,W,O).

11 % verify dominance

13 equal(J) :- cxopt(J,_,_), equal(J,W,O) : cxopt(J,W,O).

15 chain(J1,J2) :- cxopt(J1;J2,_,_), J2 < J1,
16 not cxopt(J3,W,O) : cxopt(J3,W,O) : J2 < J3 : J3 < J1.

18 check(J2) :- cxopt(J2,_,_), not chain(J1,J2) : chain(J1,J2).
19 check(J2) :- chain(J1,J2), check(J1), equal(J1).

21 bot :- not cxopt(J,W,O) : cxopt(J,W,O).
22 bot :- check(J1), worse(J1).
23 bot :- check(J1), equal(J1), not chain(J1,J2) : chain(J1,J2).

25 % require non-existence of dominating answer set

27 :- not bot.

29 % check cardinality criteria

31 index(S,Q) :- cxopt(J,_,card), minimize(J,S), wlist(S,Q,_,_),
32 not wlist(S,Q+1,E,W) : wlist(S,Q+1,E,W).

34 count(S,W,-1, 0) :- cxopt(J,W,card), minimize(J,S).
35 count(S,W,Q+1,I) :- count(S,W,Q,I), wlist(S,Q+1,_,_).
36 count(S,W,Q+1,I+1) :- count(S,W,Q,I), wlist(S,Q+1,pos(P),W), hold(P).
37 count(S,W,Q+1,I+1) :- count(S,W,Q,I), wlist(S,Q+1,neg(N),W), not hold(N).

39 cdown(S,W,Q, I) :- count(S,W,Q,I), index(S,Q), not count(S,W,Q,I+1).
40 cdown(S,W,Q-1,I) :- cdown(S,W,Q,I), wlist(S,Q,_,_).
41 cdown(S,W,Q-1,I-1) :- cdown(S,W,Q,I), wlist(S,Q,pos(P),W), true(P), 0 <= I.
42 cdown(S,W,Q-1,I-1) :- cdown(S,W,Q,I), wlist(S,Q,neg(N),W), fail(N), 0 <= I.

44 equal(J,W,card) :- cxopt(J,W,card), minimize(J,S), cdown(S,W,-1,0).

46 worse(J) :- cxopt(J,W,card), minimize(J,S), cdown(S,W,-1,-1).

48 % check inclusion criteria

50 ndiff(pos(P)) :- cxopt(J,W,incl), minimize(J,S), wlist(S,_,pos(P),W),
51 true(P).
52 ndiff(pos(P)) :- cxopt(J,W,incl), minimize(J,S), wlist(S,_,pos(P),W),
53 not hold(P).
54 ndiff(neg(N)) :- cxopt(J,W,incl), minimize(J,S), wlist(S,_,neg(N),W),
55 fail(N).
56 ndiff(neg(N)) :- cxopt(J,W,incl), minimize(J,S), wlist(S,_,neg(N),W),
57 hold(N).

59 equal(J,W,incl) :- cxopt(J,W,incl), minimize(J,S), ndiff(E) : wlist(S,_,E,W).

61 worse(J) :- cxopt(J,W,incl), minimize(J,S), wlist(S,_,pos(P),W),
62 true(P), not hold(P).
63 worse(J) :- cxopt(J,W,incl), minimize(J,S), wlist(S,_,neg(N),W),
64 fail(N), hold(N).

Listing 5. Meta-program for complex optimization (metaO.lp) on reified logic programs.

purpose of identifying undominated answer sets, and the remainder of metaO.lp defines
equal(J,W,O) and worse(J) relative to particular optimization criteria.

The rules in Line 31–46 implement the comparison of cardinalities between a candi-
date answer set and a counterexample at a priority level J over the literals indicated by a
weight W (in facts wlist(S,Q,E,W)). To this end, we first count the number of such lit-
erals E that hold wrt the candidate answer set via the rules in Line 34–37. The outcome, i.e.,
the greatest number I such that count(S,W,Q,I) is derived at the maximum index Q of
the list S given in minimize(J,S), is identified via the rule in Line 39 and taken as the
starting value for counting down the literals that hold wrt the counterexample. Note that this
approach omits a (quadratic) comparison between the precise cardinalities obtained wrt the
candidate answer set and the counterexample. The result of counting down, accomplished
via the rules in Line 40–42, is given by the derived instances of cdown(S,W,-1,I),
where two outcomes deserve particular attention. In fact, cdown(S,W,-1,0) indicates
that at least as many literals as wrt the candidate answer set hold wrt the counterexam-
ple, so that equal(J,W,card) is derived via the rule in Line 44. On the other hand,
cdown(S,W,-1,-1) signals that more literals hold wrt the counterexample; as this
makes a Pareto improvement at priority level J impossible, we derive worse(J) via the
rule in Line 46. Finally, note that we did not directly encode minimization of cost functions
with coefficients different from 1. In view of the multiset semantics supported by gringo,
such coefficients can still be represented by including the desired number of duplicates of
a literal in an input #minimize statement.

The second optimization criterion, inclusion (indicated via cxopt(J,W,incl)), is
implemented by the rules in Line 50–64. The test for equality, attested by deriving
equal(J,W,incl) via the rule in Line 59, is accomplished by checking whether a
candidate answer set and a (comparable) counterexample agree on all involved literals E;
otherwise, ndiff(E) is not derived via the rules in Line 50–57. Furthermore, the coun-
terexample is incomparable to the candidate answer set if it includes some literal not shared
by the latter; in such a case, worse(J) is derived via the rules in Line 61–62 and 63–64.
In fact, the three ⊆1

1-minimal answer sets of Π1 (given in p1.lp), consisting of the rules
in (2)–(4) and the #minimize statement in (5) can now be computed in the following way:

gringo --reify p1.lp | gringo meta.lp metaD.lp metaO.lp \
<(echo "optimize(1,1,incl).") - | claspD 0

Observe that claspD (Drescher et al. 2008), the disjunctive extension of clasp (Gebser et al.
2007), is used for solving the proper disjunctive ground program obtained from gringo.

The third optimization criterion implemented in metaO.lp (Listing 6) relies on literal
preferences according to (Sakama and Inoue 2000), i.e., a relation `1 � `2 provided via the
user predicate prefer/2. In a nutshell, an answer set X1 is preferable to another answer
set X2, i.e., X1 � X2, if there is a preference `1 � `2 such that `1 holds wrt X1, but not
wrt X2, and `2 holds wrt X2, but not wrt X1, while `′2 � `1 or `1 � `′2 applies to every
literal `′2 that holds wrt X2, but not wrt X1. The idea of the encoding part in Listing 6 is
to derive equal(J,W,pref) if a candidate answer set X1 is preferable to a counterex-
ample X2, and worse(J) if X2 is not preferable to X1. If both equal(J,W,pref)
and worse(J) are underivable, it shows that X2 is preferable to X1, but not vice versa,
so that X1 is dominated by X2 wrt the projection of the preference relation specified via
prefer/2 to literals E qualified by minimize(J,S) and wlist(S, ,E,W). To im-
plement these checks, the rules in Line 68–92 identify literals at which a candidate answer

66 % check preference criteria (relative to user predicate prefer/2)

68 cando(pos(P)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,pos(P),W),
69 fail(P), hold(P).
70 cando(neg(N)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,neg(N),W),
71 true(N), not hold(N).
72 nocan(pos(P)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,pos(P),W),
73 true(P).
74 nocan(pos(P)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,pos(P),W),
75 not hold(P).
76 nocan(neg(N)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,neg(N),W),
77 fail(N).
78 nocan(neg(N)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,neg(N),W),
79 hold(N).

81 condo(pos(P)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,pos(P),W),
82 true(P), not hold(P).
83 condo(neg(N)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,neg(N),W),
84 fail(N), hold(N).
85 nocon(pos(P)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,pos(P),W),
86 fail(P).
87 nocon(pos(P)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,pos(P),W),
88 hold(P).
89 nocon(neg(N)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,neg(N),W),
90 true(N).
91 nocon(neg(N)) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,neg(N),W),
92 not hold(N).

94 cando(S,W,E) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,E;E1,W),
95 prefer(E,E1), E1 != E, cando(E), condo(E1).

97 nocon(S,W,E) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,E,W), nocon(E).
98 nocon(S,W,E) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,E,W),
99 nocan(E1) : wlist(S,_,E1,W) : E1 != E : prefer(E,E1).

100 nocon(S,W,E) :- cxopt(J,W,pref), minimize(J,S), wlist(S,_,E;E2,W),
101 prefer(E2,E), not prefer(E,E2), cando(E2).

103 equal(J,W,pref) :- cxopt(J,W,pref), minimize(J,S), cando(S,W,E),
104 nocon(E2) : wlist(S,_,E2,W) : prefer(E2,E) : not prefer(E,E2).

106 worse(J) :- cxopt(J,W,pref), minimize(J,S), nocon(S,W,E) : wlist(S,_,E,W).

Listing 6. Meta-program for literal preferences (metaO.lp) on reified logic programs.

set X1 and a counterexample X2 differ. The literals E that hold wrt X1, but not wrt X2, are
indicated by cando(E), while nocan(E) is derived otherwise. Similarly, condo(E)
expresses that E holds wrt X2, but not wrt X1, whereas nocon(E) indicates that this is
not the case. The rule in Line 94–95 further derives cando(S,W,E) if E is a literal that
holds wrt X1 only such that E � E1 for a literal E1 that holds wrt X2 only. If such a
literal E is not defeated by another literal E2 satisfying E2 � E as well as E � E2 and
holding wrt X2 only, we derive equal(J,W,pref) via the rule in Line 103–104. On the
other hand, a literal E provides no evidence for the counterexample X2 being preferable to
the candidate answer set X1 if E does not hold wrt X2 only, if no E1 such that E � E1
holds wrt X1 only, or if some E2 such that E2 � E and E � E2 holds wrt X1 only.
If some of these cases applies, nocon(S,W,E) is derived via the rules in Line 97–101.
Given this, we further derive worse(J) via the rule in Line 106 if none of the literals E
qualified by wlist(S, ,E,W) witnesses that X2 is preferable to X1. Finally, recall that
the counterexample X2 dominates the candidate answer set X1 if X2 is preferable to X1,
but not vice versa, i.e., if neither equal(J,W,pref) nor worse(J) is derivable in
view of cxopt(J,W,pref).

For showing the correctness of our meta-programming approach, we make use of the
property that any answer set generated via meta.lp and metaD.lp encapsulates a pair
of answer sets of the reified input program.

Lemma 1
Let F be the set of facts in the reification of an extended logic program Π.

Then, X and Y are answer sets of Π iff there is a unique answer set Z of F ∪
{meta.lp} ∪ {metaD.lp} such that X = {a | hold(atom(a)) ∈ Z} and
Y = {a | true(atom(a)) ∈ Z}.

Proof (Sketch)
Given that the meta-rules in meta.lp merely reconstruct Π from its reification, we have
that X is an answer set of Π iff there is a unique answer set X ′ of F ∪ {meta.lp}
such that X = {a | hold(atom(a)) ∈ X ′}, where X ′ additionally includes F ,
atoms over hold/1 for body elements of Π that hold wrt X , as well as auxiliary
atoms over litb/1, eleb/1, and elem/1. Along with the fact that the (ground in-
stances of the) rules in metaD.lp are invariant under reduct, do not include any in-
tegrity constraint, and do not define any predicates occurring in F ∪ {meta.lp}, any
answer set X ′ of F ∪ {meta.lp} can be extended to an answer set Z = X ′ ∪ Y ′ of
F ∪ {meta.lp} ∪ {metaD.lp}, where Y ′ is a minimal model of F , the auxiliary rules
in Line 3–13 of meta.lp, and metaD.lp. In view of the (proper) disjunctive rule in
Line 10 of metaD.lp, for each a ∈ atom(Π), we have that true(atom(a)) ∈ Y ′ or
fail(atom(a)) ∈ Y ′, while saturation in Line 61–62 of metaD.lp admits bot ∈ Y ′

only if bot is necessarily derived via the rules in Line 33, 34, and 57 of metaD.lp. As
a consequence, the rules in metaD.lp check for the existence of a model Y ′ such that
Y = {a | true(atom(a)) ∈ Y ′} is an answer set of Π. In particular, the rules in
Line 33 and 34 of metaD.lp derive bot wrt interpretations that are not models of Π or
include some atom that does not occur in the head of any rule of Π whose body holds.
Moreover, the rule in Line 57 of metaD.lp derives bot wrt interpretations containing
some atom of a non-trivial SCC of the dependency graph of Π that does not belong to the
least fixpoint of T applied to the SCC. In view of Proposition 1, we conclude that Y ′ is a
minimal model (more precisely, an answer set) of F , the rules in Line 3–13 of meta.lp,
and metaD.lp such that bot /∈ Y ′ iff Y is an answer set of Π. Hence, we have that X
and Y are answer sets of Π iff there is an answer set Z of F ∪{meta.lp}∪{metaD.lp}
such that X = {a | hold(atom(a)) ∈ Z} and Y = {a | true(atom(a)) ∈ Z}.
Finally, any such answer set Z of F ∪ {meta.lp} ∪ {metaD.lp} is unique since all
atoms in Z that are not of the form hold(atom(a)) or true(atom(a)) follow de-
terministically once the interpretation of these atoms is fixed.

We are now ready to show our main correctness result wrt the specification of optimal
answer sets given in the background.

Theorem 1
Let F be the set of facts in the reification of an extended logic program Π and a statement
#minimize[`1 = w1@J1, . . . , `k = wk@Jk] such that {`1, . . . , `k} ⊆ atom(Π) ∪ {∼a |
a ∈ atom(Π)}, M ⊆ {≤wi

Ji
,⊆wi

Ji
,�wi

Ji
| 1 ≤ i ≤ k}, and � ⊆ (atom(Π) ∪ {∼a | a ∈

atom(Π)})× (atom(Π) ∪ {∼a | a ∈ atom(Π)}).

Then, X is an optimal answer set of Π wrt M iff there is a unique answer set Z of

Πmeta = F ∪ {meta.lp} ∪ {metaD.lp} ∪ {metaO.lp}
∪ {optimize(J,w,card). | ≤w

J ∈M}
∪ {optimize(J,w,incl). | ⊆w

J ∈M}
∪ {optimize(J,w,pref). | �w

J ∈M}
∪ {prefer(`′1,`′2). | `1 � `2}

such that X = {a | hold(atom(a)) ∈ Z}, where `′ = pos(atom(`)) if ` ∈
atom(Π) and, respectively, `′ = neg(atom(a)) if ` = ∼a for a ∈ atom(Π).

Proof (Sketch)
By Lemma 1, we have that X and Y are answer sets of Π iff there is a (unique) answer
set Z ′ of F ∪{meta.lp}∪{metaD.lp} such that X = {a | hold(atom(a)) ∈ Z ′}
and Y = {a | true(atom(a)) ∈ Z ′}. Since {metaO.lp} \ {:- not bot.} is
stratified, there is a unique answer set Z of Z ′ ∪ ({metaO.lp} \ {:- not bot.}) ∪
{optimize(J,w,card). | ≤w

J ∈M}∪{optimize(J,w,incl). | ⊆w
J ∈M}∪

{optimize(J,w,pref). | �w
J ∈ M} ∪ {prefer(`′1,`′2). | `1 � `2}. In the

following, we argue that, for any �wJ ∈ M and O = card/incl/pref if � = ≤/⊆/�,
we have that X�wJ Y holds iff equal(J,w,O) ∈ Z, while Y �wJ X does not hold for some
�wJ ∈M iff worse(J) ∈ Z. We thus consider the possible criteria for every �wJ ∈M :

1. If � = ≤, we have that atoms count(S,w,Q,0. . .I) are derived via the rules
in Line 7, 31–32, and 34–37 of metaO.lp, where Q is the maximum index
such that facts of the form minimize(J,S) and wlist(S,Q,`′,W) belong
to F . Then, we have that I is the cardinality of the multiset of occurrences of
` = w@J in #minimize[`1 = w1@J1, . . . , `k = wk@Jk] such that X |= `.
Starting from I in Line 39, cdown(S,w,-1,0) is derived via the rules in
Line 40–42 iff the cardinality of the multiset of occurrences of ` = w@J in
#minimize[`1 = w1@J1, . . . , `k = wk@Jk] such that Y |= ` is at least I, and
cdown(S,w,-1,-1) is derived via the same rules iff the cardinality of this mul-
tiset is greater than I. That is, X ≤w

J Y holds iff equal(J,w,card) is derived
via the rule in Line 44, and worse(J) is derived via the rule in Line 46 if Y �w

J X .
2. If � = ⊆, an atom ndiff(`′i) such that `i = w@J for some 1 ≤ i ≤ k is derived

via the rules in Line 7 and 50–57 of metaO.lp iff X 6|= `i or Y |= `i. Thus,
equal(J,w,incl) is in turn derived via the rule in Line 59 iff X ⊆w

J Y holds.
Moreover, worse(J) is derived via the rules in Line 61–62 and 63–64 if there is
some `i = w@J for 1 ≤ i ≤ k such that Y |= `i and X 6|= `i, i.e., if Y *w

J X .
3. If � = �, an atom cando(`′i), nocan(`′i), condo(`′i), or nocon(`′i) such

that `i = w@J for some 1 ≤ i ≤ k is derived via the rules in Line 7 and 68–
92 of metaO.lp iff X |= `i and Y 6|= `i, X 6|= `i or Y |= `i, X 6|= `i and
Y |= `i, or, respectively, X |= `i or Y 6|= `i. Via the rule in Line 94–95, we
further derive cando(S,w,`′) iff minimize(J,S), wlist(S, ,`′,w), and
wlist(S, ,`′1,w) belong to F such that ` � `1, X |= `, Y 6|= `, X 6|= `1,
and Y |= `1. Then, X �w

J Y holds if there is no literal `i such that `i = w@J

for some 1 ≤ i ≤ k, `i � `, ` � `i, Y |= `i, and X 6|= `i, i.e., if nocon(`′i)

holds for every `′i such that wlist(S, ,`′i,w) and prefer(`′i,`
′), but not

prefer(`′,`′i), belong to F . Hence, equal(J,w,pref) is derived via the rule
in Line 103–104 iff X �w

J Y holds. Conversely, a literal `i such that `i = w@J for
some 1 ≤ i ≤ k provides no indication for Y �w

J X if Y 6|= `i or X |= `i, if there
is no ` = w@J in #minimize[`1 = w1@J1, . . . , `k = wk@Jk] such that `i � `,
X |= `, and Y 6|= `, or if ` � `i, `i � `, X |= `, and Y 6|= ` for some ` = w@J

in #minimize[`1 = w1@J1, . . . , `k = wk@Jk]. Some of these cases applies iff
nocon(S,w,`′i) is derived via the rules in Line 97–101, so that worse(J) is in
turn derived via the rule in Line 106 if Y �w

J X .

As we have above investigated all rules that can possibly derive worse(J), i.e., the rules
in Line 46, 61–64, and 106 of metaO.lp, and since some of them applies iff �wJ ∈ M

such that Y �wJ X does not hold, we now conclude that worse(J) ∈ Z iff Y �wJ X does not
hold for some �wJ ∈ M . In addition, the above cases yield that equal(J,w,O) ∈ Z for
O = card/incl/pref iff X �wJ Y holds for � = ≤/⊆/� such that �wJ ∈M . In view of
the rules in Line 13–23, we further conclude that bot ∈ Z iff there are no priority level J
and weight w such that Y �w′

J′ X holds for all �w′

J′ ∈M where J ′ ≥ J and X �wJ Y does not
hold for �wJ ∈M . That is, bot ∈ Z iff X is an answer set of Π that is not dominated by the
answer set Y of Π wrt M . However, for any answer set Z of Πmeta, saturation in Line 61–62
of metaD.lp admits bot ∈ Z only if bot is derived wrt every answer set Y of Π, i.e., if
there is no answer set Y of Π that dominates X wrt M . Hence, the integrity constraint in
Line 27 of metaO.lp selects exactly the answer sets Z of Πmeta\{:- not bot.} such
that X = {a | hold(atom(a)) ∈ Z} is an undominated and thus optimal answer set
of Π wrt M . Finally, any such answer set Z of Πmeta is unique since all atoms in Z that are
not of the form hold(atom(a)) follow deterministically once the interpretation of these
atoms is fixed. In particular, bot and true(atom(a)) as well as fail(atom(a)) for
all a ∈ atom(Π) hold in view of the integrity constraint in Line 27 of metaO.lp along
with saturation in Line 61–62 of metaD.lp.

Regarding the computational complexity of tasks that can be addressed using our meta-
programming approach to optimization, we first note that deciding whether there is an
optimal answer set is in NP , as the existence of some answer set (decidable by means of
meta.lp only) is sufficient for concluding that there is also an optimal one. However, the
inherent complexity becomes more sensible if we consider the question of whether some
atom a belongs to an optimal answer set. To decide it, one can augment the reified input
program (but not the input program itself), meta.lp, metaD.lp, and metaO.lp with
the integrity constraint :- not hold(atom(a)). Then, several complex optimization
criteria at a single priority level 1 lead to completeness for ΣP

2 , the second level of the
polynomial time hierarchy, thus showing that disjunctive ASP is appropriate to implement
them. To see this, note that deciding whether an atom a belongs to some answer set of a
positive disjunctive logic program is ΣP

2 -complete (Eiter and Gottlob 1995). When dis-
junctions a1 ∨ · · · ∨ ak in the heads of rules are rewritten to 1 #sum[a1 = 1, . . . , ak = 1],
the question of whether an atom a belongs to an answer set of the original program can
be addressed by reifying the rewritten program, adding the integrity constraint :- not
hold(atom(a)), and applying meta.lp, metaD.lp, and metaO.lpwrt several op-
timization criteria. For one, we can include a #minimize statement over all atoms of the



Complex Optimization in Answer Set Programming — Extended Version — 19

input program, each associated with a different weight, to exploit the Pareto improvement
implemented in metaO.lp for refuting a candidate answer set including a if it does not
correspond to a minimal model, i.e., an answer set of the original program. Alternatively,
we can include a #minimize statement over all atoms of the input program, each hav-
ing the weight 1, and augment the meta-program with the fact optimize(1,1,incl).
We could also use a #minimize statement over all atoms ai of the input program along
with their negation, each having the weight 1, and add the facts optimize(1,1,pref)
as well as prefer(neg(atom(ai)),pos(atom(ai))). In view of these reductions,
we conclude that Pareto efficiency, inclusion, and literal preferences independently capture
computational tasks located at the second level of the polynomial time hierarchy, and our
meta-programs allow for addressing them via an extended program along with facts (and
possibly also integrity constraints) steering optimization relative to its reification.

5 Applications: A Case Study

While the approach of Eiter and Polleres (2006) consists of combining two separate logic
programs, one for “guessing” and a second one for “checking,” into a disjunctive program
addressing both tasks, our meta-programming technique applies to a single (reified) input
program along with complex optimization criteria. In fact, we provide a generic implemen-
tation of such criteria on top of extended programs encoding solution spaces. Hence, our
meta-programming technique allows for a convenient representation of reasoning tasks in
which testing the optimality of solutions to an underlying problem in NP lifts the complex-
ity to ΣP

2 -hardness. Respective formalisms include ordinary, parallel, as well as prioritized
circumscription (McCarthy 1980; Lifschitz 1985), minimal consistency-based diagnosis
(Reiter 1987), and preferred extensions of argumentation frameworks (Dung 1995). Simi-
larly, Pareto efficiency is an important optimality condition in decision making (Chevaleyre
et al. 2007) and system design (Gries 2004). In the following, we illustrate the application
of our approach on the example of an existing real-world application: repair wrt large
gene-regulatory networks (Gebser et al. 2010).

Listing 7 shows a simplified version of the repair encoding given in (Gebser et al. 2010).
It applies to a regulatory network, a directed graph with (partially) labeled edges, repre-
sented by facts of the predicates vertex/1, edge/2, and obs elabel/3, where a la-
bel S is 1 (activation) or -1 (inhibition). In addition, the data of experiments labeled P are
provided by facts of the predicates exp/1, inp/2 denoting input vertices (subject to per-
turbations), and obs vlabel/3, where a label S is again 1 (increase) or -1 (decrease).
The regulatory network is consistent with the experiment data if there are total labelings of
edges and vertices (for each experiment P) such that the label of every non-input vertex V
is explained by the influence of some of its regulators U, where the influence is the product
S*T of the edge label S and the label T of U (in experiment P). In the practice of systems
biology, regulatory networks and experiment data are often collected from heterogeneous
sources, and likewise incomplete or noisy. In view of this, it is very likely that a regulatory
network and experiment data are mutually inconsistent, which makes it highly non-trivial
to draw biologically meaningful conclusions in an automated way. To address this short-
age, several repair operations were devised in (Gebser et al. 2010), which can be enabled
via facts of the form repair(K,J,W), where K indicates a certain kind of admissible

1 % auxiliary concepts

3 sign(-1;1).
4 complement(S,-S) :- sign(S).

6 % construct candidate repair

8 pos(aedge(U,V), J,W) :- repair(aedge,J,W), vertex(U;V), U != V.
9 pos(eflip(U,V,S),J,W) :- repair(eflip,J,W), obs_elabel(U,V,S).

10 pos(ivert(V), J,W) :- repair(ivert,J,W), vertex(V).
11 pos(pvert(P,V), J,W) :- repair(pvert,J,W), exp(P), vertex(V).
12 pos(vflip(P,V,S),J,W) :- repair(vflip,J,W), obs_vlabel(P,V,S).

14 { apply(R) } :- pos(R,_,_).

16 % construct consistent total labelings

18 elabel(U,V,S) :- not apply(eflip(U,V,S)), obs_elabel(U,V,S).
19 elabel(U,V,T) :- apply(eflip(U,V,S)), complement(S,T).
20 elabel(U,V,S) :- apply(aedge(U,V)), not elabel(U,V,T), complement(S,T).
21 elabel(U,V,S) :- edge(U,V), not elabel(U,V,T), complement(S,T).

23 vlabel(P,V,S) :- not apply(vflip(P,V,S)), obs_vlabel(P,V,S).
24 vlabel(P,V,T) :- apply(vflip(P,V,S)), complement(S,T).
25 vlabel(P,V,S) :- not vlabel(P,V,T), complement(S,T), exp(P), vertex(V).

27 inf(P,V,S*T) :- elabel(U,V,S), vlabel(P,U,T), not inp(P,V).

29 :- vlabel(P,V,S), not inf(P,V,S), not inp(P,V), not apply(ivert(V);pvert(P,V)).

31 % optimize repair

33 #minimize [ apply(R) = W @ J : pos(R,J,W) ].

Listing 7. Encoding of repair wrt regulatory networks and experiment data (repair.lp).

repair operations, J a priority level, and W a weight. The repair operations R to apply are
selected via the rule in Line 14 of Listing 7, and their effects are propagated via the rules in
Line 18–29, thus obtaining total edge and vertex labelings witnessing the reestablishment
of consistency. Given that applications of repair operations modify a regulatory network or
experiment data (depending on the kind of operations), we are interested in applying few
operations only, which is expressed by the #minimize statement in Line 33.

A reasonable repair configuration could consist of facts of the following form:

repair(ivert,J1,W1). admitting to turn vertices into inputs in all experiments.
repair(eflip,J2,W2). admitting network modifications by flipping edge labels.
repair(pvert,J3,W3). admitting to turn vertices into inputs in specific experiments.
repair(vflip,J4,W4). admitting data modifications by flipping vertex labels.

While the kinds of repair referred to by ivert and eflip operate primarily on a network
(in view of incompleteness or incorrectness), the ones denoted by pvert and vflip
mainly address the data (which can be noisy). If we penalize all repair operations uni-
formly via J = J1 = J2 = J3 = J4 and W = W1 = W2 = W3 = W4, the instantiation of the
#minimize statement in Line 33 represents ordinary cardinality-based optimization, as-
sembled in solvers like clasp and smodels. However, by adding optimize(J,W,incl)
as a fact, we can easily switch to inclusion-based minimization and use a disjunctive solver
like claspD to solve the more complex problem. While our meta-programs enable such
a shift of optimization criteria by means of adding just one fact, a direct disjunctive en-
coding of inclusion-based minimization has been provided in (Gebser et al. 2010); note

that the latter is by far more involved than the basic repair encoding in Listing 7. Further-
more, our meta-programming approach allows us to distinguish between different kinds of
repair operations (without prioritizing them) and optimize wrt Pareto efficiency. To accom-
plish this, one only needs to pick unequal values for W1, . . . ,W4, where cardinality-based
minimization wrt each Wi can selectively be replaced by inclusion via providing a fact
optimize(J,Wi,incl). Finally, we can choose to rank kinds of repair operations by
providing different priority levels J1, . . . ,J4. For instance, network repairs may be consid-
ered as “more drastic” operations than data repairs, which can be expressed by assigning
levels such that J1 = J2 > J3 = J4. In fact, the kinds of admissible repair operations
as well as appropriate penalties for applying them depend on the biological application
and experience. In this respect, the flexibility gained due to meta-programming allows for
deploying and comparing different optimization criteria, e.g., regarding the accuracy of
resulting predictions (cf. (Gebser et al. 2010)).

For giving an account of the practical capabilities of our meta-programming approach,
we empirically compared it to the direct encoding of inclusion-based minimization in (Geb-
ser et al. 2010). To this end, we ran gringo version 3.0.3 and claspD version 1.1 on 100
instances (random samples from real data of two biological experiments, as also used in
(Gebser et al. 2010)) wrt three kinds of admissible repair operations, resulting in 300 runs
each with our meta-programs and with the direct encoding. All runs have been performed
sequentially on a machine equipped with Intel Xeon E5520 processors and 48 GB main
memory under Linux, imposing a time limit of 4000 sec per run. (In view of moderate
memory consumption, a space limit was not needed.) To our own surprise, more runs
were completed in time with the meta-programs than with the direct encoding: 219 ver-
sus 150.4 The disadvantages of the direct encoding show that further gearing would be
required to improve solving efficiency, which adds to the difficulty of furnishing a func-
tional saturation-based encoding. In view of this, we conclude that our meta-programming
approach to complex optimization is an eligible and viable alternative. However, enhance-
ments of disjunctive ASP solvers boosting its performance would still be desirable.

6 Discussion

The major contribution of this work is the provision of easy modeling capacities for ex-
pressing complex preferences involving a higher computational complexity. We accom-
plish this by developing advanced meta-modeling techniques that allow us to reinterpret
#minimize statements by means of ASP. This methodology offers a generic saturation-
based implementation technique for testing various optimality conditions wrt answer sets
of extended logic programs, as used with lparse and gringo. At (metasp), we provide en-
hanced versions of our meta-encodings, supporting additional features of gringo’s input
language (Gebser et al.) as well as some basic optimizations with regard to grounding per-
formance. These encodings provide a ready-to-use platform for implementing applications
involving complex optimization criteria in ASP.

Our integral approach to modeling complex optimization criteria in ASP brings about a

4 All instances and detailed results are available at (metasp).

number of individual contributions. To begin with, we introduce the reification capacities
of our grounder gringo along with the associated meta-encoding, paving the way to the
immediate availability of meta-modeling techniques. In fact, the full version of the basic
meta-encoding in Listing 1 covers the complete language of gringo, including disjunc-
tions and diverse aggregates. Moreover, our meta-modeling techniques provide a general
account of saturation and, thus, abolish its compulsory replication for expressing complex
preferences. Of particular interest is the stability property of answer sets serving as implicit
counterexamples. Unlike the approach of Eiter and Polleres (2006), our encoding avoids
“guessing” level mappings. Also, our target language involves choice rules and #sum con-
straints (Simons et al. 2002), and we are unaware of any pre-existing meta-encoding of
corresponding answer sets, neither as candidates nor as counterexamples. Likewise, related
meta-programming approaches for generating consequences of logic programs (Faber and
Woltran 2009) or explanations wrt debugging queries (Oetsch et al. 2010) do not consider
such aggregates (but disjunctions in object programs).

We exploit the two-dimensionality of #minimize statements by using levels and
weights for combining a lexicographic ranking with Pareto efficiency. At each level, one
may choose whether groups of literals sharing the same weight are compared wrt cardinal-
ity or inclusion. Also, summation can be modeled by exploiting the multiset property of
#minimize statements. This is extended by the framework of (Sakama and Inoue 2000),
relying on a preference relation among literals (given in addition to #minimize state-
ments). In fact, the approach of Section 4 allows for capturing the special cases of parallel
and prioritized circumscription, investigated by Janhunen and Oikarinen in (2004; 2008).
An interesting future extension is the encoding of optimality conditions for logic programs
with ordered disjunction (Brewka et al. 2004), whose custom-made implementation in the
prototype psmodels interleaves two smodels oracles for accomplishing a generate-and-test
approach similar to the idea of our meta-programs. Ultimately, our approach could serve
as an implementation platform for answer set optimization (Brewka et al. 2003) and the
preference description language proposed in (Brewka 2004). Last but not least, our meta-
programs furnish a rich and readily available source of ΣP

2 -hard challenge problems, fos-
tering the development of ASP solvers for disjunctive logic programming.

